

コインの転がり運動解析

～ Maple によるDAE問題の数値解析 ～

本ワークシートでは、Maple でのDAE (Differential Algebraic Equation) 問題の数値解析事例として「コインの転がり運動」を示します。

Maple には、DAE 問題に対応した数値計算ソルバが実装されており、数式及び初期値を定義することで計算することが可能です。DAE 問題の例として、コインの転がり運動を題材とし、マルチボディダイナミクスの考え方に基づいた数式展開を行った後に二つのケースで数値計算し、その結果を示します。




図1：コインの座標系

<参考文献>

- [1] 社団法人 日本機械学会 (2007)
マルチボディダイナミクス(1) - 基礎理論 - コロナ社
- [2] 社団法人 日本機械学会 (2007)
マルチボディダイナミクス(2) - 数値解析と実際 - コロナ社

ワークシートの初期化及びパッケージの呼び出し

ワークシートの初期化を行います。

```
> restart;
```

次に、使用する三つのパッケージを呼び出します。パッケージの詳細については以下を参照ください。

- [LinearAlgebra](#)
- [VectorCalculus](#)
- [plots](#)

```
> with( LinearAlgebra );  
> with( VectorCalculus );  
> with( plots );
```

解析モデルの定義

図 1 に示したコインのモデル図より求めた速度の拘束式を示し、Lagrange 乗数を用いた運動方程式を求めます。

速度の拘束式と一般化座標、一般化速度の定義

図 1 より速度の拘束式は次式と求まります (途中計算、省略)。

```
> VelConsteq := [  
  diff(X(t), t) = R * diff(theta(t), t) * sin(phi(t)) * sin(theta  
    (t)) - R * (diff(phi(t), t) * cos(theta(t)) + diff(psi(t), t))  
    * cos(phi(t)),  
  diff(Y(t), t) = -R * diff(theta(t), t) * cos(phi(t)) * sin(theta  
    (t)) - R * (diff(phi(t), t) * cos(theta(t)) + diff(psi(t), t))  
    * sin(phi(t)),  
  diff(Z(t), t) = R * diff(theta(t), t) * cos(theta(t)) ]:  
map(print, VelConsteq);
```

$$\begin{aligned}\dot{X}(t) &= R \dot{\theta}(t) \sin(\phi(t)) \sin(\theta(t)) - R (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t) \cos(\phi(t))) \\ \dot{Y}(t) &= -R \dot{\theta}(t) \cos(\phi(t)) \sin(\theta(t)) - R (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t) \sin(\phi(t))) \\ \dot{Z}(t) &= R \dot{\theta}(t) \cos(\theta(t))\end{aligned}\quad [] \quad (2.1.1)$$

次に、一般化座標は、以下であるため、一般化速度で表すことができます。

```
> Pos := Matrix(5, 1, [X(t), Y(t), phi(t), theta(t), psi(t)]  
) :  
Pos^%T;
```

$$\begin{bmatrix} X(t) & Y(t) & \phi(t) & \theta(t) & \psi(t) \end{bmatrix} \quad (2.1.2)$$

一般化速度 :

```
> Vel := Jacobian( convert(Pos,list), [t]):
Vel^%T;
```

$$\begin{bmatrix} \dot{X}(t) & \dot{Y}(t) & \dot{\phi}(t) & \dot{\theta}(t) & \dot{\psi}(t) \end{bmatrix} \quad (2.1.3)$$

運動方程式の導出

Lagrange 乗数を用いた運動方程式を導出します。
 まず、運動エネルギー T は、並進運動エネルギーと回転運動エネルギーの和であるため、コインの質量 m 、各軸のイナーシャを I_x 、 I_y 、 I_z とすると以下になります。

```
> eqT := 1/2 * m * ( diff( X(t), t)^2 + diff( Y(t), t)^2 +
diff( Z(t), t)^2 ) +
1/2 * ( Ix * omegax(t)^2 + Iy * omegay(t)^2 + Iz *
omegaz(t)^2 );
eqT;
```

$$\frac{m(\dot{X}(t)^2 + \dot{Y}(t)^2 + \dot{Z}(t)^2)}{2} + \frac{I_x \text{omegax}(t)^2}{2} + \frac{I_y \text{omegay}(t)^2}{2} + \frac{I_z \text{omegaz}(t)^2}{2} \quad (2.2.1)$$

また、(2.2.1) 式中の回転角速度成分は、図 1 より次式となります。

```
> omegaeq := [ omegax(t) = diff(theta(t),t),
omegay(t) = diff(phi(t),t) * sin(theta(t)),
omegaz(t) = diff(phi(t), t) * cos(theta(t)) +
diff(psi(t), t) ];
map(print, omegaeq);
```

$$\begin{aligned} \text{omegax}(t) &= \dot{\theta}(t) \\ \text{omegay}(t) &= \dot{\phi}(t) \sin(\theta(t)) \\ \text{omegaz}(t) &= \dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t) \end{aligned} \quad (2.2.2)$$

したがって、(2.1.1) 式の $Z(t)$ の微分に関する速度拘束式と、(2.2.1) 式及び (2.2.2) 式より、運動エネルギー T の式は以下と整理することができます。

```
> eqT2 := subs( [ omegaeq[], VelConsteq[3] ], eqT );
eqT2;
```

$$\begin{aligned} &\frac{m(\dot{X}(t)^2 + \dot{Y}(t)^2 + R^2 \dot{\theta}(t)^2 \cos^2(\theta(t)))}{2} + \frac{I_x \dot{\theta}(t)^2}{2} + \frac{I_y \dot{\phi}(t)^2 \sin^2(\theta(t))}{2} \\ &+ \frac{I_z (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t))^2}{2} \end{aligned} \quad (2.2.3)$$

次に、コインの半径を R とすると、回転半径 κ は、次式となります。

```
> rot := kappa = R / sqrt(2):
rot;
```

$$\kappa = \frac{R\sqrt{2}}{2} \quad (2.2.4)$$

また、コインが均一で真円であると仮定すると、各イナーシャ I_x 、 I_y 、 I_z は、以下となります。

```

> inertiaeq := [ Ix = m * kappa^2 / 2,
                 Iy = m * kappa^2 / 2,
                 Iz = m * kappa^2      ]:
map(print, inertiaeq);

```

$$I_x = \frac{m \kappa^2}{2}$$

$$I_y = \frac{m \kappa^2}{2}$$

$$I_z = m \kappa^2$$

(2.2.5)

ここで、ポテンシャルエネルギー V は、以下になります。

```

> eqV := m * g * R * sin(theta(t)):
eqV;

```

$$m g R \sin(\theta(t))$$

(2.2.6)

そのため、ラグランジェ関数 $L=T-V$ は次式と求まります。

```

> eqL := subs( inertiaeq, eqT2 - eqV):
eqL;

```

$$\frac{m (\dot{X}(t)^2 + \dot{Y}(t)^2 + R^2 \dot{\theta}(t)^2 \cos^2(\theta(t)))}{2} + \frac{m \kappa^2 \dot{\theta}(t)^2}{4} + \frac{m \kappa^2 \dot{\phi}(t)^2 \sin^2(\theta(t))}{4}$$

$$+ \frac{m \kappa^2 (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t))^2}{2} - m g R \sin(\theta(t))$$

(2.2.7)

また、この後、式操作を行い易くするために、一般化座標及び一般化速度の変数を次のように置き換えます。

```

> PosTrans := [ Pos[1,1] = X, Pos[2,1] = Y,
                Pos[3,1] = phi, Pos[4,1] = theta, Pos[5,1] =
                psi ]:
map(print, PosTrans);

```

$$X(t) = X$$

$$Y(t) = Y$$

$$\phi(t) = \phi$$

$$\theta(t) = \theta$$

$$\psi(t) = \psi$$

(2.2.8)

```

> VelTrans := [ Vel[1,1] = dX, Vel[2,1] = dY,
                Vel[3,1] = dphi, Vel[4,1] = dtheta, Vel[5,1]
                = dpsi ]:
map(print, VelTrans);

```

$$\dot{X}(t) = dX$$

$$\dot{Y}(t) = dY$$

$$\dot{\phi}(t) = dphi$$

$$\dot{\theta}(t) = dtheta$$

$$\dot{\psi}(t) = dpsi$$

[] (2.2.9)

したがって、ラグランジェ関数Lは次式となります。

```
> eqL2 := subs( [PosTrans[], VelTrans[]], eqL):
eqL2;
```

$$\frac{m(dX^2 + dY^2 + R^2 d\theta^2 \cos(\theta)^2)}{2} + \frac{m\kappa^2 d\theta^2}{4} + \frac{m\kappa^2 d\phi^2 \sin(\theta)^2}{4} + \frac{m\kappa^2 (d\phi \cos(\theta) + d\psi)^2}{2} - mgR \sin(\theta)$$

(2.2.10)

質量マトリクスを求めます。
ラグランジェ関数Lの一般化速度によるヤコビアンを求めます。

```
> diffeqL := simplify([ diff( eqL2, dX), diff( eqL2, dY),
diff( eqL2, dphi), diff( eqL2,
dtheta), diff( eqL2, dpsi)]):
map(print, diffeqL);
```

$$\begin{matrix} m dX \\ m dY \\ \frac{m\kappa^2 (d\phi \cos(\theta)^2 + 2 \cos(\theta) d\psi + d\phi)}{2} \\ \frac{m d\theta (2 R^2 \cos(\theta)^2 + \kappa^2)}{2} \\ m\kappa^2 (d\phi \cos(\theta) + d\psi) \end{matrix}$$

[] (2.2.11)

よって、(2.2.11)式より質量マトリクスは次のようになります。

```
> Mtemp1, Mtemp2 := GenerateMatrix(diffeqL, [dX, dY, dphi,
dtheta, dpsi]):
Mass := subs([ seq( rhs(PosTrans[i]) = lhs(PosTrans[i]), i
= 1..nops(PosTrans) ) ],
Mtemp1):
Mass;
```

$$\begin{bmatrix} m & 0 & 0 & 0 & 0 \\ 0 & m & 0 & 0 & 0 \\ 0 & 0 & \frac{m\kappa^2 (\cos(\theta(t))^2 + 1)}{2} & 0 & m\kappa^2 \cos(\theta(t)) \\ 0 & 0 & 0 & \frac{m(2R^2 \cos(\theta(t))^2 + \kappa^2)}{2} & 0 \\ 0 & 0 & m\kappa^2 \cos(\theta(t)) & 0 & m\kappa^2 \end{bmatrix}$$

(2.2.12)

さらに、運動方程式を立てるために、速度拘束式のヤコビアン、外力、Lagrange 乗数を定義します。

速度拘束式のヤコビアン :

```
> VelJac, Vtemp := GenerateMatrix( [ VelConsteq[1],
  VelConsteq[2] ], convert( Vel, list)):
  VelJac;
```

$$\begin{bmatrix} 1 & 0 & R \cos(\theta(t)) \cos(\phi(t)) & -R \sin(\phi(t)) \sin(\theta(t)) & R \cos(\phi(t)) \\ 0 & 1 & R \cos(\theta(t)) \sin(\phi(t)) & R \cos(\phi(t)) \sin(\theta(t)) & R \sin(\phi(t)) \end{bmatrix} \quad (2.2.13)$$

外力 :

```
> Force := Matrix(5,1,
  [ 0,
    0,
    m * kappa^2 * (diff(phi(t),t) * diff(theta(t),
  t) * cos(theta(t)) * sin(theta(t)) + diff(psi(t),t) * diff(theta
  (t),t) * sin(theta(t))),
  -m * ( (1/2 * kappa^2 * diff(phi(t),t)^2 - R^2 *
  diff(theta(t),t)^2) * cos(theta(t)) * sin(theta(t)) + kappa^2 *
  diff(phi(t),t) * diff(psi(t),t) * sin(theta(t)) + g * R * cos(theta
  (t))),
  m * kappa^2 * diff(phi(t),t) * diff(theta(t),
  t) * sin(theta(t)) ]);
```

$$Force := \begin{bmatrix} 0 \\ 0 \\ m \kappa^2 (\dot{\phi}(t) \dot{\theta}(t) \cos(\theta(t)) \sin(\theta(t)) + \dot{\psi}(t) \dot{\theta}(t) \sin(\theta(t))) \\ -m \left(\left(\frac{\kappa^2 \dot{\phi}(t)^2}{2} - R^2 \dot{\theta}(t)^2 \right) \cos(\theta(t)) \sin(\theta(t)) + \kappa^2 \dot{\phi}(t) \dot{\psi}(t) \sin(\theta(t)) + g R \cos(\theta(t)) \right) \\ m \kappa^2 \dot{\phi}(t) \dot{\theta}(t) \sin(\theta(t)) \end{bmatrix} \quad (2.2.14)$$

Lagrange乗数 :

```
> LagMulti := Matrix(2,1, [lambda1(t), lambda2(t)]);
```

$$LagMulti := \begin{bmatrix} \lambda_1(t) \\ \lambda_2(t) \end{bmatrix} \quad (2.2.15)$$

以上より、Lagrange 乗数を用いた運動方程式は次式と求まります。

```
> AccTrans := [seq( cat(d, rhs(VelTrans[i])) = diff(Vel[i,
  1],t),
  i = 1..nops(VelTrans))]:
  Dynamicseqsub1 := GenerateEquations(Mass, [seq( lhs
```

```
(AccTrans[i]), i=1..nops(AccTrans)),
      convert(Force, Vector)):
Dynamiceqsub2 := GenerateEquations(VelJac^%T, [lambda1,
lambda2]):
Dynamiceq0 := [ seq( Dynamiceqsub1[i] + Dynamiceqsub2[i],
i=1..nops(Dynamiceqsub1)) ]:
Dynamiceq := subs([AccTrans[], lambda1 = LagMulti[1,1],
lambda2 = LagMulti[2,1] ],
      Dynamiceq0):
map(print,Dynamiceq);
```

$$m\ddot{X}(t) + \lambda 1(t) = 0$$

$$m\ddot{Y}(t) + \lambda 2(t) = 0$$

$$\frac{m\kappa^2 \ddot{\phi}(t) \cos(\theta(t))^2}{2} + \frac{m\kappa^2 \ddot{\phi}(t)}{2} + m\kappa^2 \cos(\theta(t)) \ddot{\psi}(t)$$

$$+ R \cos(\theta(t)) \cos(\phi(t)) \lambda 1(t) + R \cos(\theta(t)) \sin(\phi(t)) \lambda 2(t) = m\kappa^2 (\dot{\phi}(t) \dot{\theta}(t) \cos(\theta(t)) \sin(\theta(t)) + \dot{\psi}(t) \dot{\theta}(t) \sin(\theta(t)))$$

$$\frac{m(2R^2 \cos(\theta(t))^2 + \kappa^2) \ddot{\theta}(t)}{2} - R \sin(\phi(t)) \sin(\theta(t)) \lambda 1(t)$$

$$+ R \cos(\phi(t)) \sin(\theta(t)) \lambda 2(t) = -m \left(\left(\frac{\kappa^2 \dot{\phi}(t)^2}{2} - R^2 \dot{\theta}(t)^2 \right) \cos(\theta(t)) \sin(\theta(t)) + \kappa^2 \dot{\phi}(t) \dot{\psi}(t) \sin(\theta(t)) + g R \cos(\theta(t)) \right)$$

$$m\kappa^2 \cos(\theta(t)) \ddot{\phi}(t) + m\kappa^2 \ddot{\psi}(t) + R \cos(\phi(t)) \lambda 1(t) + R \sin(\phi(t)) \lambda 2(t) = m\kappa^2 \dot{\phi}(t) \dot{\theta}(t) \sin(\theta(t))$$

[]

(2.2.16)

Maple による DAE 問題の数値解析

前セクションで定義した「コインの転がりモデル」の運動方程式を利用して、Maple による DAE 問題の数値解析を行います。

パラメータ定義

パラメータ以下のように定義します。

R : コインの半径
 m : コインの質量
 κ : コインの回転半径
 ρ : コインの中心円軌道の半径
 g : 重力

```
> params := [ R = 0.25, m = 1, kappa = 0.25/sqrt(2), rho =
5*0.25, g = 9.81 ]:
map(print, params);
```

$R=0.25$

$$\begin{aligned}
 m &= 1 \\
 \kappa &= 0.1250000000 \sqrt{2} \\
 \rho &= 1.25 \\
 g &= 9.81 \\
 &[]
 \end{aligned}
 \tag{3.1.1}$$

▼ ケース 1 : 定常歳差運動

ケース 1 として、定常歳差運動をする場合の解析を行います。
 まず、一般化座標の各初期値を以下とします。

```

> IniPos1 := [X0 = 0, Y0 = -5 * 0.25, phi0 = 0, theta0 =
  Pi/3, psi0 = 0]:
  map(print, IniPos1);

```

$$\begin{aligned}
 X0 &= 0 \\
 Y0 &= -1.25 \\
 \phi0 &= 0 \\
 \theta &= \frac{\pi}{3} \\
 \psi0 &= 0 \\
 &[]
 \end{aligned}
 \tag{3.2.1}$$

次に一般化速度の各初期値を計算します。歳差角速度の初期値は以下となります。

```

> IniVel1 := [0,0,0,0,0]:
  IniVel1[3] := dphi0 = evalf(subs([params[],IniPos1[]],
  sqrt( 4*g*cot(theta0)/(R*cos(theta0)+6*rho))))):
  IniVel1[3];

```

$$dphi0 = 1.723710241
 \tag{3.2.2}$$

さらに、スピン角速度及び章動角速度の初期値は次式とします。

```

> IniVel1[4] := dtheta0 = 0:
  IniVel1[4];
  IniVel1[5] := dpsi0 = evalf(subs([params[],IniPos1[],
  IniVel1[3]],-(rho/R + cos(theta0))*dphi0)):
  IniVel1[5];

```

$$\begin{aligned}
 dtheta0 &= 0 \\
 dpsi0 &= -9.480406326
 \end{aligned}
 \tag{3.2.3}$$

また、X及びY方向の速度初期値は、速度拘束式より次のように求められます。

```

> IniVel1sub := subs( [ params[], seq(lhs(PosTrans[i]) = cat
  ( rhs(PosTrans[i]), 0), i =3..5),
  seq(lhs(VelTrans[i]) = cat
  ( rhs(VelTrans[i]), 0), i =3..5) ],
  rhs( VelConsteq[1] ) ):
  IniVel1[1] := dX0 = evalf(subs([IniPos1[], IniVel1[3..5][
  ], IniVel1sub)):
  IniVel1[1];
  IniVel2sub := subs( [ params[], seq(lhs(PosTrans[i]) = cat

```



```
( rhs(PosTrans[i]), 0), i = 3..5),
      seq( lhs(VelTrans[i]) = cat
( rhs(VelTrans[i]), 0), i = 3..5) ],
      rhs( VelConsteq[2] ) ):
IniVel1[2] := dY0 = evalf(subs([IniPos1[], IniVel1[3..5][
], IniVel2sub)):
IniVel1[2];
      dX0=2.154637801
      dY0=0.
(3.2.4)
```

パラメータ定義及び上記で定義した初期値を使用して、運動方程式及び速度拘束式を整理すると次式となります。

```
> Transeq := [seq( diff( lhs( VelTrans[i] ), t) = diff(
apply( rhs(VelTrans[i] ), t), t), i=1..5)]:
deq1sub := subs(Transeq, [ Dynamiceq[], VelConsteq[1..2][
] )):
deq1 := [subs(params, deq1sub)[], seq( apply( rhs(VelTrans
[i]), t) = Vel[i,1], i = 1..5)]:
map(print, deq1);
```

$$\dot{d}X(t) + \lambda_1(t) = 0$$

$$\dot{d}Y(t) + \lambda_2(t) = 0$$

$$0.01562500000 \dot{d}\phi(t) \cos(\theta(t))^2 + 0.01562500000 \dot{d}\phi(t) \\ + 0.03125000000 \cos(\theta(t)) \dot{d}\psi(t) + 0.25 \cos(\theta(t)) \cos(\phi(t)) \lambda_1(t) \\ + 0.25 \cos(\theta(t)) \sin(\phi(t)) \lambda_2(t) = 0.03125000000 \dot{\phi}(t) \\ \dot{\theta}(t) \cos(\theta(t)) \sin(\theta(t)) + 0.03125000000 \dot{\psi}(t) \dot{\theta}(t) \sin(\theta(t)) \\ \frac{(0.1250 \cos(\theta(t))^2 + 0.03125000000) \dot{d}\theta(t)}{2} \\ - 0.25 \sin(\phi(t)) \sin(\theta(t)) \lambda_1(t) + 0.25 \cos(\phi(t)) \sin(\theta(t)) \lambda_2(t) = \\ - (0.01562500000 \dot{\phi}(t)^2 - 0.0625 \dot{\theta}(t)^2) \cos(\theta(t)) \sin(\theta(t)) \\ - 0.03125000000 \dot{\phi}(t) \dot{\psi}(t) \sin(\theta(t)) - 2.4525 \cos(\theta(t)) \\ 0.03125000000 \cos(\theta(t)) \dot{d}\phi(t) + 0.03125000000 \dot{d}\psi(t) + 0.25 \cos(\phi(t)) \lambda_1(t) \\ + 0.25 \sin(\phi(t)) \lambda_2(t) = 0.03125000000 \dot{\phi}(t) \dot{\theta}(t) \sin(\theta(t)) \\ \dot{X}(t) = 0.25 \dot{\theta}(t) \sin(\phi(t)) \sin(\theta(t)) - 0.25 (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t)) \cos(\phi(t)) \\ \dot{Y}(t) = -0.25 \dot{\theta}(t) \cos(\phi(t)) \sin(\theta(t)) - 0.25 (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t)) \sin(\phi(t)) \\ dX(t) = \dot{X}(t) \\ dY(t) = \dot{Y}(t) \\ d\phi(t) = \dot{\phi}(t) \\ d\theta(t) = \dot{\theta}(t) \\ d\psi(t) = \dot{\psi}(t) \\ [] \quad (3.2.5)$$

上記より、(3.2.5)式を使用し、dsolve関数にて数値解析を行います。
数値計算ソルバは、DAE問題であるため、rkf45_daeを使用します。dsolve関数の
詳細については、[dsolve](#)をご参照ください。

```
> Inieq1 := [ seq( subs( [ t = 0 ], [ Pos[i,1] = rhs(IniPos1
  [ i ] ),
                                apply(rhs(VelTrans[i]),
  t) = rhs(IniVel1[i]) ) ) [ ], i=1..5) ] :
sol1 := dsolve( [ deq1[ ], Inieq1[ ] ], numeric);


sol1 := proc(x_rkf45_dae) ... end proc (3.2.6)
```

数値計算結果をプロットし、確認します。


```
> p1_1_1 := odeplot( sol1, [ t, X(t) ], 0..30, numpoints=
200, color=red):
p1_1_2 := odeplot( sol1, [ t, Y(t) ], 0..30, numpoints=
200, color=blue):
p1_1 := display([ p1_1_1, p1_1_2 ], title="コインの重心Gの位
置変化(X(t), Y(t))"):


p1_2_1 := odeplot( sol1, [ t, phi(t) ], 0..30, numpoints=
200, color=red):
p1_2_2 := odeplot( sol1, [ t, theta(t) ], 0..30,
numpoints=200, color=blue):
p1_2 := display([ p1_2_1, p1_2_2 ], title="歳差角phiと章動角
thetaの変化" ):

p1_3 := odeplot( sol1, [ X(t), Y(t) ], 0..90, numpoints=
500, title="コインの重心Gの軌道" ):
display( Vector( [p1_1, p1_2, p1_3] ) );
```



歳差角phiと章動角thetaの変化





▼ ケース 2 : ふらつきを伴う転がり運動

ケース 2 として、ふらつきを伴う転がり運動をする場合の解析を行います。
まず、一般化座標の各初期値を以下とします。

```
> IniPos2 := [X0 = 0, Y0 = 0, phi0 = 0, theta0 = Pi/2-0.001,
psi0 = 0]:
map(print, IniPos2);
```

$X0 = 0$

$Y0 = 0$

$\phi0 = 0$

$\theta0 = \frac{\pi}{2} - 0.001$

$\psi0 = 0$

[]

(3.3.1)

次に一般化速度の各初期値を計算します。歳差角速度の初期値は以下となります。

```
> IniVel2 := [0,0,0,0,0]:
IniVel2[3] := dphi0 = 0:
IniVel2[3];
```

$$dphi0=0 \quad (3.3.2)$$

さらに、スピン角速度及び章動角速度の初期値は次式とします。

```
> IniVel2[4] := dtheta0 = 0:
IniVel2[4];
IniVel2[5] := dpsi0 = evalf(subs([params[],IniPos2[],
IniVel2[3]],0.2*sqrt(g*R/3)/R)):
IniVel2[5];
```

$$dtheta0=0$$

$$dpsi0=0.7233256529 \quad (3.3.3)$$

また、X及びY方向の速度初期値は、速度拘束式より次のように求められます。

```
> IniVel1sub := subs( [ params[], seq(lhs(PosTrans[i]) = cat
( rhs(PosTrans[i]), 0), i =3..5),
seq(lhs(VelTrans[i]) = cat
( rhs(VelTrans[i]), 0), i =3..5) ],
rhs( VelConsteq[1] ) ):
IniVel2[1] := dX0 = evalf(subs([IniPos2[], IniVel2[3..5][
], IniVel1sub)):
IniVel2[1];
IniVel2sub := subs( [ params[], seq(lhs(PosTrans[i]) = cat
( rhs(PosTrans[i]), 0), i =3..5),
seq(lhs(VelTrans[i]) = cat
( rhs(VelTrans[i]), 0), i =3..5) ],
rhs( VelConsteq[2] ) ):
IniVel2[2] := dY0 = evalf(subs([IniPos2[], IniVel2[3..5][
], IniVel2sub)):
IniVel2[2];
```

$$dX0=-0.1808314132$$

$$dY0=-0. \quad (3.3.4)$$

パラメータ定義及び上記で定義した初期値を使用して運動方程式及び速度拘束式を整理すると次式となります。

```
> Transeq := [seq( diff( lhs( VelTrans[i] ), t) = diff(
apply( rhs(VelTrans[i] ), t), t), i=1..5)]:
deq2sub := subs(Transeq, [ Dynamiceq[], VelConsteq[1..2][
] )):
deq2 := [subs(params, deq2sub)[], seq( apply( rhs(VelTrans
[i]), t) = Vel[i,1], i = 1..5)]:
map(print, deq2);
```

$$\dot{d}X(t) + \lambda_1(t) = 0$$

$$\dot{d}Y(t) + \lambda_2(t) = 0$$

$$0.01562500000 \dot{d}phi(t) \cos(\theta(t))^2 + 0.01562500000 \dot{d}phi(t)$$

$$+ 0.03125000000 \cos(\theta(t)) \dot{d}psi(t) + 0.25 \cos(\theta(t)) \cos(\phi(t)) \lambda_1(t)$$

$$+ 0.25 \cos(\theta(t)) \sin(\phi(t)) \lambda_2(t) = 0.03125000000 \dot{\phi}(t)$$

$$\dot{\theta}(t) \cos(\theta(t)) \sin(\theta(t)) + 0.03125000000 \dot{\psi}(t) \dot{\theta}(t) \sin(\theta(t))$$

$$\begin{aligned}
& \frac{(0.1250 \cos(\theta(t))^2 + 0.03125000000) d\theta(t)}{2} \\
& - 0.25 \sin(\phi(t)) \sin(\theta(t)) \lambda_1(t) + 0.25 \cos(\phi(t)) \sin(\theta(t)) \lambda_2(t) = \\
& - (0.01562500000 \dot{\phi}(t)^2 - 0.0625 \dot{\theta}(t)^2) \cos(\theta(t)) \sin(\theta(t)) \\
& - 0.03125000000 \dot{\phi}(t) \dot{\psi}(t) \sin(\theta(t)) - 2.4525 \cos(\theta(t)) \\
& 0.03125000000 \cos(\theta(t)) d\phi(t) + 0.03125000000 d\psi(t) + 0.25 \cos(\phi(t)) \lambda_1(t) \\
& + 0.25 \sin(\phi(t)) \lambda_2(t) = 0.03125000000 \dot{\phi}(t) \dot{\theta}(t) \sin(\theta(t)) \\
& \dot{X}(t) = 0.25 \dot{\theta}(t) \sin(\phi(t)) \sin(\theta(t)) - 0.25 (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t)) \cos(\phi(t)) \\
& \dot{Y}(t) = -0.25 \dot{\theta}(t) \cos(\phi(t)) \sin(\theta(t)) - 0.25 (\dot{\phi}(t) \cos(\theta(t)) + \dot{\psi}(t)) \sin(\phi(t)) \\
& dX(t) = \dot{X}(t) \\
& dY(t) = \dot{Y}(t) \\
& d\phi(t) = \dot{\phi}(t) \\
& d\theta(t) = \dot{\theta}(t) \\
& d\psi(t) = \dot{\psi}(t)
\end{aligned} \tag{3.3.5}$$

よって、(3.3.5)式を使用し、dsolve関数にて数値解析を行います。
ただし、本ケースでは、問題がstiffであるため、ソルバとして、rosenbrock_daeを使用します。

```

> Inieq2 := [ seq( subs( [ t = 0 ], [ Pos[i,1] = rhs(IniPos2
[i ]),
                                apply(rhs(VelTrans[i]),
t) = rhs(IniVel2[i]) )][], i=1..5) ]:
sol2 := dsolve( [ deq2[], Inieq2[] ], numeric, method =
rosenbrock_dae, maxfun = 500000);
sol2 := proc(x_rosenbrock_dae) ... end proc

```

数値計算結果をプロットし、確認します。

```


> p2_1_1 := odeplot( sol2, [ t, X(t) ], 0..30, numpoints=
200, color=red):
p2_1_2 := odeplot( sol2, [ t, Y(t) ], 0..30, numpoints=
200, color=blue):
p2_1 := display([ p2_1_1, p2_1_2 ], title="コインの重心Gの位
置変化(X(t), Y(t))"):


p2_2_1 := odeplot( sol2, [ t, phi(t) ], 0..30, numpoints=
200, color=red):
p2_2_2 := odeplot( sol2, [ t, theta(t) ], 0..30,
numpoints=200, color=blue):
p2_2 := display([ p2_2_1, p2_2_2 ], title="歳差角phiと章動角
thetaの変化"):


p2_3 := odeplot( sol2, [ X(t), Y(t) ], 0..90, numpoints=
500, title="コインの重心Gの軌道" ):
display( Vector( [p2_1, p2_2, p2_3] ) );

```

コインの重心Gの位置変化(X(t), Y(t))







▼ まとめ

本ワークシートでは、コインの転がりモデルを例題として、MapleによるDAE問題の数値解析の一手法を示しました。dsolve関数は、DAE形式の方程式をそのまま入力することで数値解析を行うことができるDAEソルバを持つため非常に効率良く解析を行うことができます。

また、本ワークシートでは示していませんが、アニメーション機能を活用することでMaple内でコインの転がりを可視化することも出来ますし、MapleSimと併用することで様々な方法で使用することができます。